Garage Door Monitor

Installed 1/6/2018

The monitor provides a visual indication to the driver of the car's clearance relative to the door. The beam obstruction detector for the door has been mounted at the car's bumper height. When the door is not fully opened (e.g. while the door is opening, or when the door has stopped partly open), the indicator is RED. When the door is fully open and there is clearance to shut the door, the indicator is GREEN. When the door is fully open but the car is obstructing closure of the door (e.g. the car is driving into the garage) the indicator is orange. So when arriving at the garage and the door begins to open, the indicator is RED (car should wait). Once the door is fully open and the indicator turns GREEN, the car may proceed into the garage. As the car is entering, the light will be ORANGE to indicate that the door is not yet clear. When there is sufficient clearance to close the door, the indicator turns GREEN. And once the door begins to close, the indicator will again be RED.

To provide power for the monitor, a +12V "wall wart" power pack is mounted near the door-opener motor and gets AC power from the same receptacle as the motor. A four-wire cable using the BLACK for Ground and the YELLOW for +12V Power runs to the monitor. The door-opener track has a switch which is activated by the door reaching a selected position on the track. The door-opener uses this input to stop the motor. This high-impedance signal level is brought to the monitor on the RED wire. The GREEN wire is tied to the Beam Obstruction Detection circuit (the wires from the two units across the door-opening).



Figure 1: Beam Obstruction Detection Signal

The signal from the beam is not a simple on/off. It has a pulse output (see Figure 1) with a minimum of 2.5 Volts, and a maximum of 13 Volts. If there is no voltage, the door-opener detects an error. If the beam is interrupted, the output is just the 2.5 Volt level. When the beam is clear, the pulses occur at 325 Hz.



Figure 2: Beam Signal Input to Processor

To use the signal as an input to the processor, it must be reduced to roughly the 0V to 5V range. A capacitor input was used to remove the DC offset, a zener diode to limit the high to +5V and a silicon diode to limit the undershoot. Because this circuit creates a decay characteristic to the signal, the software triggers on the (cleaner) rising edge.



Figure 3: Garage Door Monitor Schematic

The track-switch level is monitored on the RED wire. It is on/off at about +12V (door not fully open to 0 volt (and zero ohms) when the door is fully opened. The 100 K resistor was selected by experiment - it provides sufficient input to the processor, does not interfere with the door-opener.

The "superbright" LED's were measured for their forward voltage drop. Red was 1.75 volts, Orange was 2.05 volts, and Green was 2.7 volts. Series resistors were calculated for 15 mA current, and standard resistor values close to the calculated values were selected.

For redundancy, there are two red LED's on independent pins. The primary LED on pin 10 (C0) is constant-on when selected. The secondary red LED on pin 7 (C3) is pulsed at 40 Hz, so has a lower series resistance to roughly equalize the perceived brightness between the two red LED's. The 40 Hz frequency produces a blink-rate not perceived by the eye's center of vision, but is perceived by the peripheral vision. If you look at it, you can't see it. You have to look past it to see it - intended to help grab someone's attention.

The physical construction is in a metal enclosure. The (four wire "telephone") cable is run open across the ceiling to the North garage wall, and in metal wire-mold down to the enclosure. The enclosure is mounted at a driver's eye level, determined by experiment, which provides good visibility both forward and rearward (for backing into the garage).

Appendix A: Assembly Code Listing

```
;======================  Garage Door Monitor ======================
;                  Roger Himka - Release V1.0 - 1/4/2018
;
;     The Garage Door obstruction monitor (Infra-red beam) produces pulses
;     at 325 Hz with a pulse width of about 2.5 milli-seconds and 1.0 ms low
;     when the beam is clear of obstructions.  High is about 13 V and low is
;     about 2.5 V.  When the beam is obstructed, the pulses stop.
;
;     The "door closed sense switch" is a high impedance 12 V which becomes about 9 V
;     with a 100 K Ohm load.  The signal is HIGH when the door is not fully closed.
;
;     The display logic is RED if the door is not fully opened.  If the "door closed
;     sense switch" input to A5 is low, the door is fully open.  The beam status
;     is determined by counting pulses input to A2 for 0.0125 ms (1/80 th of a second).
;     If there are pulses (there should be three or four), the door is clear and GREEN
;     is displayed.  Otherwise, if there are no pulses, the beam is obstructed and
;     ORANGE is displayed.
;
;     The RED display consists of two red LEDs, one of which is a continuous display
;     and the other is pulsed at 40 Hz to enhance it's attention-getting.
;
;=====================================================================
;
;     TIMER2 is used to generate the 80 Hz beam sample period, and then the 40 Hz
;     enable for the flashing red LED. TIMER0 is input from A2 (constrained to 0 to
;     +5 V range by diodes) and counts the beam pulses for one sample period.  The TIMER2
;     is just used in a delay loop with the interrupt flag monitored by software in the
loop.
;
;=====================================================================

#include <p16F684.inc>
      __config (_INTRC_OSC_NOCLKOUT & _WDT_OFF & _PWRTE_OFF & _MCLRE_OFF & _CP_OFF
                  & _IESO_OFF & _FCMEN_OFF)

     cblock 0x20
Beam_Status
Door_Status
PortCContents
RED_40Hz_LEDStatus
     endc

;   --- INPUTS ---
#define OBSTRUCTION_DETECTION_BEAM  2     ; A2, Pin 11      Infra-red beam pulses
                                          ;                 (325 Hz when beam is clear)
#define DOOR_STATUS_SWITCH          5     ; A5, Pin 2       Switch input is HIGH
                                          ;                 when door is not fully open

;   --- Beam Status ---
#define     BEAM_CLEAR              0
#define     BEAM_OBSTRUCTED         1

;   --- Door Status ---
#define DOOR_OPEN 0
#define DOOR_CLOSED     1

;   --- OUTPUTS ---
#define RED_LED         0     ; C0, Pin 10
#define ORANGE_LED      1     ; C1, Pin 9
#define GREEN_LED       2     ; C2, Pin 8
#define RED_40Hz_LED    3     ; C3, Pin 7
```

```
      org 0
Start:
;=====================================================
      nop
      nop

;-------- INITIALIZATIONS --------

      ;-------- BEGIN Register Bank 0 Initialization --------

      ;     Bank 0,     Set up TIMER2 for 0.0125 second interval (80 Hz)
      ;     Set up for 0.0125 second interval (40 Hz square wave)
      movlw b'00100111' ; Postscaler=3 0010, Timer2=On 1, Prescaler=16 11
      movwf T2CON

      ;-------- END Register Bank 0 Initialization --------

      ;-------- BEGIN Register Bank 1 Initialization --------
      bsf         STATUS,RP0        ; Select Register Bank 1

      ;     Set up TIMER0 to count pulses from the IR beam b'X0111XXX'
            ;  For Bank1, bit 6 INTEDG =0 (falling edge), bit 5 T0CS =1 (external input),
            ;                       bit 4 T0SE =0 (inc on low-to-high), bit 3 PSA =1
            ;                       (prescaler assigned to WDT, removed from TMR0),
            ;                        bits 2=0 PS<2:0> =XXX
      movlw b'00110000'       ;     Configure Timer0.   Sourced from external
                                                            (IR beam detector)
      movwf OPTION_REG        ;

      ;     Bank 1,     Set up TIMER2 for 0.0125 second interval
      movlw 09B   ; Value in comparison register - with prescaler 1:3
                              provides 0.0125 seconds (40 Hz square wave)
      movwf PR2

      clrf  TRISC             ; Make PortC all outputs

      clrf  ANSEL             ; Make PortA all digital

      movlw   b'00100100'           ;
      movwf   TRISA           ; Make PortA A2 and A5 inputs

      bsf         OPTION_REG,NOT_RAPU     ; Disable pull-up enabling
                              ; Pull-ups provided in hardware where needed

      bcf     STATUS,RP0      ; Un-Select Register Bank 1 (Select Bank 0)
      ;-------- END Register Bank 1 Initialization --------

      ;-------- Hardware Initialization --------
      clrf  PORTC                   ; All outputs off
      clrf  PortCContents           ; All status to off
      clrf  RED_40Hz_LEDStatus      ;

;=============================================================
;-------- Main Program --------

MonitorLoop:

      clrf  TMR0        ; Clear the beam pulse count
      bcf   PIR1,TMR2IF ; Clear the TIMER2 interrupt flag
Timer2Loop:             ; Use TIMER2 to wait 0.0125 seconds while T0
                        ; counts beam pulses (three to four expected)
      btfss PIR1,TMR2IF
      goto  Timer2Loop
      ; TIMER2 has rolled over
EndTimer2Loop:
```

```
BeamCheck:  ; Get the number of IR pulses from TIMER0
      movf  TMR0,w
      btfss STATUS,Z
      goto  GotPulses
NoPulses:
      movlw BEAM_OBSTRUCTED
      movwf Beam_Status
      goto  EndBeamCheck
GotPulses:
      movlw BEAM_CLEAR
      movwf Beam_Status
EndBeamCheck:

DoorCheck:
      btfsc PORTA,DOOR_STATUS_SWITCH
      goto  DoorClosed
DoorOpen:
      movlw DOOR_OPEN
      movwf Door_Status
      goto  EndDoorCheck
DoorClosed:
      movlw DOOR_CLOSED
      movwf Door_Status
EndDoorCheck:

SelectLED:
      movlw       DOOR_OPEN
      xorwf Door_Status,w
      btfss STATUS,Z
      goto  DoorIsNotOpen
DoorIsOpen:
      movlw BEAM_CLEAR
      xorwf Beam_Status,w
      btfss STATUS,Z
      goto BeamIsObstructed
      BeamIsClear:
            movlw b'00000100'       ; GREEN_LED
            goto  EndSelectLED
      BeamIsObstructed:
            movlw b'00000010'       ; ORANGE LED
            goto  EndSelectLED
DoorIsNotOpen:
      movlw b'00000001'       ; RED_LED
      iorwf RED_40Hz_LEDStatus,w ; Include RED 40Hz LED every second period (40 Hz blink)
      movwf PortCContents
EndSelectLED:
      movwf PORTC


Begin40HzLED:
      movf  RED_40Hz_LEDStatus,w
      btfss STATUS,Z
      goto  ClearRed40HzLED
SetRed40HzLED:
      movlw b'00001000'
      movwf RED_40Hz_LEDStatus
      goto  End40HzRed
ClearRed40HzLED
      clrf  RED_40Hz_LEDStatus
End40HzRed:

      goto  MonitorLoop

;=========================================================
      end
```